7-25-2024

Using Excel's "LAMBDA" Feature to Produce Multi-Stage Stock Pricing Functions for the Classroom

Tom Arnold

Joseph Farizo

Jonathan Godbey

Using Excel's "LAMBDA" Feature to Produce Multi-Stage Stock Pricing Functions for the Classroom

Tom Arnold, CFA, CIPM
The Robins School of Business
University of Richmond
Department of Finance
102 UR Drive
Richmond, VA 23173
tarnold@richmond.edu
O: 804-287-6399
F: 804-289-8878

Joseph Farizo
The Robins School of Business
Department of Finance
102 UR Drive
University of Richmond, VA 23173
O: 804-289-8565
F: 804-289-8878
jfarizo@richmond.edu

Jonathan M. Godbey
Robinson College of Business
Georgia State University
Department of Finance
33 Gilmer Street SE
Atlanta, GA 30303
Oxford, OH 45056-1879
jgodbey@gsu.edu
O: 404-413-7328
F: 404-413-7312

July 25, 2024

Using Excel's "LAMBDA" Feature to Produce Multi-Stage Stock Pricing Functions for the Classroom

Excel's new "=LAMBDA" function is applied to multi-stage stock pricing. This provides a context for implementing the =LAMBDA function, a greater understanding of the concept of the time value of money, and a reduction in the complexity of performing the calculation. The =LAMBDA functions lessens the need for using VBA and macros and only requires basic knowledge of Excel function syntax.

INTRODUCTION

The newest version of Excel has the "=LAMBDA" function that allows the programmer to create functions that do not require the Visual Basic for Applications (VBA) environment nor macros. For example, using common stock price inputs, such as, current dividend (D0), dividend growth rate (g), and discount rate (k), the programmer can "test" a =LAMBDA function for pricing a common stock in this manner:

=LAMBDA(D0,g,k,D0*(1+g)/(k-g))(cell assigned to "D0", cell assigned to "g", cell assigned to "k") to replicate the common stock pricing formula:

$$\text{Stock price} = D0 \times (1 + g) \div (k - g) \quad (1)$$

The interior of the =LAMBDA function, defines three parameters: "D0," "g," and "k," followed by a formula to price the common stock using the listed parameters. The cell assignments in parentheses after the function, identify within the spreadsheet where the parameters: "D0," "g," and "k" are located. Note: the LAMBDA function can have any number of parameters, but can only have one formula defined using the listed parameters.

Once tested, the =LAMBDA function can be named and requires three inputs: =STOCKPRICED0(D0,g,k). The =STOCKPRICED0 function can be implemented anywhere within the spreadsheet as:

=STOCKPRICED0(cell assigned to "D0", cell assigned to "g", cell assigned to "k").

In the next section, the steps for creating the =STOCKPRICED0 function are demonstrated. In section two, the =STOCKPRICED0 function is implemented within two-stage and three-stage stock pricing formulas using the techniques of Alexander, Arnold, and Wu (2021). In section three, classroom applications are suggested with section four concluding the article.

SECTION 1: Defining the =STOCKPRICED0 Function

In Figure 1, the parameters for pricing a common stock are presented within an Excel spreadsheet along with the calculation of the common stock price.

Figure 1: Implementing =LAMBDA and =STOCKPRICED0 Functions

	A	B	C	D
1	**Common Stock:**			
2				
3	Current Dividend:	$ 0.45		
4	Dividend Growth Rate:	6.00%		
5	Discount Rate:	10.00%		
6				
7	Stock Price:	$ 11.9250		
8				

Cell B7 with a calculation: =B3*(1 + B4) / (B5 – B4)
Cell B7 with a LAMBDA function: =LAMBDA(D0,g,k,D0*(1+g)/(k-g))(B3, B4, B5)
Cell B7 with STOCKPRICED0 function: = STOCKPRICED0(B3, B4, B5) created using: =LAMBDA(D0,g,k,D0*(1+g)/(k-g))

A copy of this spreadsheet is available at: https://scholarship.richmond.edu/finance-faculty-publications/XX/

The calculation of the common stock price can be performed using a traditional Excel calculation in cell B7:

=B3*(1 + B4) / (B5 – B6)

To test the LAMBDA function, cell B7 becomes:

=LAMBDA(D0,g,k,D0*(1+g)/(k-g))(B3, B4, B5)

To create the =STOCKPRICED0 function, the following steps are required (note: the previous LAMBDA function does not need to exist in the spreadsheet to create the STOCKPRICED0 function):

- Go to the “Formulas” menu
- Click the “Name Manager” symbol
- Click “New” in the menu that appears

- In the next menu that appears
    - Name: type STOCKPRICED0
    - Scope: Workbook (this is the default setting)
    - Comment: type in description of the function if desired
    - Refers to: type =LAMBDA(D0,g,k,D0*(1+g)/(k-g))
    - Click "OK"

The Excel file now has new function: = STOCKPRICED0 that requires three parameters in a set order: "D0," "g," and "k." With the new function defined, cell B7 becomes: =STOCKPRICED0(B3, B4, B5). If the cell containing the =STOCKPRICED0 function is copied into a second Excel file, the =STOCKPRICED0 function will become part of the functions available within that second Excel file in addition to the being available in the original file.

SECTION 2: 2-Stage and 3-Stage Stock Pricing Functions: STOCKPRICEG1 and STOCKPRICEG2

For the two-stage stock price example, define: D0 = \$0.45, k = 10.00%, GROW1 = 15.00%, GROWD1 = 5, and g = 6.00%. Compared to the previous example, two new parameters emerge: GROW1 = 15.00% is the dividend growth rate for the next five periods (i.e. GROWD1 or "growth duration" equals 5 periods). After five periods of 15% growth, the dividend growth rate returns to its normal level of 6.00% (i.e. "g"). Following, Alexander, Arnold, and Wu (2021), define the following:

$$K_1 = (1 + k) / (1 + \text{GROW1}) \quad (2)$$

$$k_1 = K_1 - 1 \quad (3)$$

$$\text{Stock price} = \frac{D0}{k_1}\left[1 - \frac{1}{(1+k_1)^{\text{GROWD1}}}\right] + \frac{[D0\times(1+g)/(k-g)]}{(1+k_1)^{\text{GROWD1}}} \quad (4)$$

Essentially, the stock price becomes the same as bond pricing with the yield-to-maturity being "k_1", the coupon being D0, the maturity being GROWD1, and the PAR payment being the stock price without considering the effect of GROW1.

Again, follow the process for creating a new function to calculate the two-stage stock price:

- Go to the "Formulas" menu
- Click the "Name Manager" symbol
- Click "New" in the menu that appears
- In the next menu that appears
    - Name: type STOCKPRICEG1
    - Scope: Workbook (this is the default setting)
    - Comment: type in description of the function if desired
    - Refers to: type =LAMBDA(D0,GROW1,GROWD1,g,k,PV((1 + k) / (1 + GROW1) – 1, GROWD1, -D0, -STOCKPRICED0(D0,g,k)))
    - Click "OK"

As demonstrated in equation (4), the =STOCKPRICEG1 function is a bond pricing calculation, which in Excel can be calculated with the =PV function:

=PV(yield-to-maturity, maturity, - coupon, - PAR)

Negative values for the coupon and the PAR are adjustments for Excel's default setting of cash flows being negative. For the two-stage stock price: the yield-to-maturity is: (1 + k) / (1 + GROW1) – 1, the maturity is GROWD1, the coupon is D0, and the PAR is D0 × (1 + g) ÷ (k – g) or STOCKPRICED0(D0,k,g).

From a programming perspective, the student can see that the =LAMBDA function can incorporate existing Excel functions, such as =PV, and Excel functions created by the programmer, the =STOCKPRICED0 function (see Figure 2).

Figure 2: Implementing the =STOCKPRICEG1 Function

	A	B	C	D
1	**Common Stock:**			
2				
3	Current Dividend:	$ 0.45	D0	
4	Dividend Growth Rate:	6.00%	g	
5	Discount Rate:	10.00%	k	
6				
7	Stock Price:	$ 11.9250		
8				
9	**Two-Stage Dividend Growth:**			
10				
11	Current Dividend:	$ 0.45	D0	
12	Dividend Growth Rate 1:	15.00%	GROW1	
13	Growth Rate Duration 1:	5	GROWD1	
14	Perpetual Dividend Growth Rate:	6.00%	g	
15	Discount Rate:	10.00%	k	
16				
17	Stock Price:	$ 17.4691		

Cell B7: = STOCKPRICED0(B3, B4, B5) created using:
=LAMBDA(D0,g,k,D0*(1+g)/(k-g))

Cell B17: = STOCKPRICEG1(B11, B12, B13, B14, B15) created using:
=LAMBDA(D0,GROW1,GROWD1,g,k,PV((1 + k)/(1 + GROW1) – 1, GROWD1, -D0, -STOCKPRICED0(D0,g,k)))

A copy of this spreadsheet is available at: https://scholarship.richmond.edu/finance-faculty-publications/XX/

The calculation by equation is:

$K_1 = (1 + 10.00\%) / (1 + 15.00\%) = 0.9565217$ (5)

$k_1 = K_1 - 1 = -4.34783\%$ (6)

$$\text{Stock price} = \frac{\$0.45}{-4.34783\%}\left[1 - \frac{1}{(1+(-4.34783\%))^5}\right] + \frac{[\$0.45\times(1+6.00\%)/(10.00\%-6.00\%)]}{(1+(-4.34783\%))^5}$$

Stock price = $17.4691 (7)

From a pedagogy perspective (as demonstrated in Alexander, Arnold, and Wu, 2021), there is a benefit to breaking down a complicated calculation into a bond pricing equation that can be implemented into Excel or a financial calculator. In an iterative fashion, a three-stage stock price follows the same logic of becoming the calculation of a bond price.

A three-stage stock pricing example incorporates a third different dividend growth rate of 10.00% (GROW2) for a duration of 5 periods (GROWD2). In a similar manner as to how the =STOCKPRICED0 function is nested within the =STOCKPRICEG1 function, the =STOCKPRICEG1 function will be nested within a new function =STOCKPRICEG2 for calculating a 3-stage stock price.

First, clarify what is happening with the dividend growth:

- D0 = \$0.45, which grows 15.00% (GROW1) for five (GROWD1) periods
- The dividend will then grow at 10.00% (GROW2) for another five (GROWD2) periods
- After ten periods of abnormal growth, the dividend will grow at 6.00% (g) perpetually
- The appropriate discount rate is 10.00% (k)

To calculate the 3-stage stock price, determine the stock price as if only GROW2 and GROWD2 applies:

$K_1 = (1 + 10.00\%) / (1 + 10.00\%) = 1.00000$ (8)

$k_1 = K_1 - 1 = 0.00\%$ (9)

$$\text{Stock price} = \frac{\$0.45}{0.00\%}\left[1 - \frac{1}{(1+(0.00\%))^5}\right] + \frac{[\$0.45\times(1+6.00\%)/(10.00\%-6.00\%)]}{(1+(0.00\%))^5}$$

$$\text{Stock price} = 5 \times \$0.45 + \frac{[\$0.45\times(1+6.00\%)/(10.00\%-6.00\%)]}{(1)^5}$$

Stock price = $14.1750 (10)

Next, consider the calculation in equation (10) as the PAR value for a bond with D0 as the coupon, the yield-to-maturity based on GROW1 and k, and a maturity of GROWD1:

$K_1 = (1 + 10.00\%) / (1 + 15.00\%) = 0.9565217$ (11)

$k_1 = K_1 - 1 = -4.34783\%$ (12)

$$\text{Stock price} = \frac{\$0.45}{-4.34783\%}\left[1 - \frac{1}{(1+(-4.34783\%))^5}\right] + \frac{\$14.1750}{(1+(-4.34783\%))^5}$$

Stock price = $20.2791 (13)

Equation (13) is the solution for the three-stage stock price and required two bond pricing iterations: equations (8) through (10) and equations (11) through (13). In Excel, the iterative process can be incorporated into a function:

= STOCKPRICEG2(D0,GROW1,GROWD1,GROW2,GROWD2,g,k)

The new function will incorporate the =STOCKPRICEG1 function, which in tun, incorporates the =STOCKPRICED0 function (see Figure 3).

Figure 3: Implementing the =STOCKPRICEG2 Function

	A	B	C	D
1	**Common Stock:**			
2				
3	Current Dividend:	$ 0.45	D0	
4	Dividend Growth Rate:	6.00%	g	
5	Discount Rate:	10.00%	k	
6				
7	Stock Price:	$ 11.9250		
8				
9	**Two-Stage Dividend Growth:**			
10				
11	Current Dividend:	$ 0.45	D0	
12	Dividend Growth Rate 1:	15.00%	GROW1	
13	Growth Rate Duration 1:	5	GROWD1	
14	Perpetual Dividend Growth Rate:	6.00%	g	
15	Discount Rate:	10.00%	k	
16				
17	Stock Price:	$ 17.4691		

18				
19	**Three-Stage Dividend Growth:**			
20				
21	Current Dividend:	$ 0.45	D0	
22	Dividend Growth Rate 1:	15.00%	GROW1	
23	Growth Rate Duration 1:	5	GROWD1	
24	Dividend Growth Rate 2:	10.00%	GROW2	
25	Growth Rate Duration 2:	5	GROWD2	
26	Perpetual Dividend Growth Rate:	6.00%	g	
27	Discount Rate:	10.00%	k	
28				
29	Stock Price:	$ 20.2791		

Cell B7: = STOCKPRICED0(B3, B4, B5) created using:
=LAMBDA(D0,g,k,D0*(1+g)/(k-g))

Cell B17: = STOCKPRICEG1(B11, B12, B13, B14, B15) created using:
=LAMBDA(D0,GROW1,GROWD1,g,k,PV((1 + k)/(1 + GROW1) – 1, GROWD1, -D0, -STOCKPRICED0(D0,g,k)))

Cell B29: = STOCKPRICEG2(B21, B22, B23, B24, B25, B26, B27) created using:
=LAMBDA(D0,GROW1,GROWD1,GROW2, GROWD2,g,k,PV((1 + k)/(1 + GROW1) – 1, GROWD1, -D0, -STOCKPRICEG1(D0,GROW2,GROWD2,g,k)))

A copy of this spreadsheet is available at: https://scholarship.richmond.edu/finance-faculty-publications/XX/

The process for creating a new function to calculate the three-stage stock price is:

- Go to the "Formulas" menu
- Click the "Name Manager" symbol
- Click "New" in the menu that appears
- In the next menu that appears
    - Name: type STOCKPRICEG2
    - Scope: Workbook (this is the default setting)
    - Comment: type in description of the function if desired
    - Refers to: type

      =LAMBDA(D0,GROW1,GROWD1,GROW2,GROWD2,g,k,

      PV((1 + k) / (1 + GROW1) – 1, GROWD1, -D0,

      -STOCKPRICEG1(D0,GROW2,GROWD2,g,k)))

- Click “OK”

Following the nesting process of the stock pricing functions can lead to valuations for N-stage stock prices. The associated =STOCKPRICEGX function, where X = N – 1 will still be a bond pricing equation with the PAR being =STOCKPRICEGY, where Y = N – 2. This type of iterative programming can form the basis for teaching DO-loops in other programming languages or one can assign a challenging problem, such as, a four-stage stock price in Excel after creating the functions in Figure 3 in a class. Alternatively, one can perform the two-stage stock pricing function in class and assign a three-stage stock pricing problem.

SECTION 3: Classroom Applications

Although some applications have already been presented, a more formalized approach is presented in this section.

Finance classroom:

There is value outside of using Excel to demonstrate how to break down a two-stage stock price into a bond price equation. It requires a knowledge of how a bond price is calculated and really emphasizes time value of money skills. Further, the idea of a terminal value in a corporate valuation follows from this logic particularly when the coupons become the periodic cash flows generated from a firm.

Excel in the Finance Classroom:

Although there is interest in finance students to learn other programming languages, Excel is still a dominant skill for employers. Ability using and understanding the new =LAMBDA feature will certainly impress in an interview and will also be a very useful skill going forward. By copying a cell containing a user-created function into a second

Excel file, the second file will now have the user-created function (and any nested user-created functions) within the function menu. It will not be long before sets of user-created functions become available for a price or readily downloaded from web resources. Consequently, understanding and using the =LAMBDA function will be very advantageous for the student and will require greater emphasis time value of money skills.

Programming:

Perhaps not in a finance classroom, but in a computer science or information science classroom, the nesting property demonstrated with N-stage stock prices becomes the context for programming a DO-loop, rather than an exercise for summing the numbers from one to one hundred. By demonstrating the nesting property through repetition of newly created =STOCKPRICEGX type functions, the necessity for a DO-loop becomes emphasized and for a reasonably small DO-loop, a means for checking if the Do-loop is working correctly.

SECTION 4: Conclusion

The newly available =LAMBDA function in Excel creates many possibilities for the classroom. By applying it to multi-stage stock pricing, it provides a context for understanding the mathematics required for pricing the stock and it provides richer insights into the concept of the time value of money, which often gets lost as students memorize formulas while foregoing context. Knowledge of the new =LAMBDA function will certainly impress employers during interviews and only requires knowledge of Excel rather than a programming language. As students and instructors become more comfortable with the =LAMBDA function, many new applications of complex calculations will emerge.

REFERENCES

Alexander, Maura, Tom Arnold, and Ge Wu. 2021. "Multi-Stage Stock Pricing Techniques for the Classroom." *Journal of Economics and Finance Education*, Volume 20, Number 2, pp. 14-21.